
*=============EDGE_Detection=====================================
When one want to do anything using discreet time, having things
happen on two different clock edges can save a lot of trouble. 

======================================================================
Edge detection is usually done with a gate time delay and 
another gate. This edge detector just uses an R and C and
some math. 

*=========Rising_Edge=========================
.SUBCKT   RISE_E VIN    OUT  
 R1       VIN    VTD    1meg
 C1       VTD    0      1p 
 BRE      OUT    0      V = 5*u(V(VIN)-V(VTD)-.1) 
.ENDS     RISE_E
*  ___                         ___
* |VIN|_/\  /\  /\_ VTD    ___|OUT|
* |___|   \/  \/  _|_    _|_  |___|
*              C1 ___   /BRE\
*       R1         |    \___/
*                 _|_    _|_
*                 ///    ///
======================================================================
It is desirable to be able to put out separate pulses when 
a clock signal goes high and when it goes low. 

*=========Rise_and_Fall_SubCircuit============================================
Rising_Edge
*     ________________________
*    _|_     _|__           __|_
*   /_  \   |    |  ___    |    |  ___
*  // \  \  |RISE|_|VRE|   |FALL|_|VFE|
*  \   \//  |____| |___|   |____| |___|
*   \___/
*    _|_  VIN
*    ///
*
*V_PULSE# NODE_P NODE_N DC     VALUE  PULSE( VINIT  VPULSE TDELAY TRISE  TFALL  PWIDTH PERIOD )
VIN       VIN    0      DC     0      PULSE( 0      5      1n     1n     1n     50u     100u   )
XRE       VIN    VRE    RISE_E
XFE       VIN    VFE    FALL_E

.control 
*TRAN     TSTEP  TSTOP  TSTART TMAX   
tran      1u     3m     0       1u
set       pensize = 2
plot      vin vre vfe
s.endc
======================================================================

===========Full_Netlist_For_Copy_Paste=======================
Rising_Edge
*     ________________________
*    _|_     _|__           __|_
*   /_  \   |    |  ___    |    |  ___
*  // \  \  |RISE|_|VRE|   |FALL|_|VFE|
*  \   \//  |____| |___|   |____| |___|
*   \___/
*    _|_  VIN
*    ///
*



*V_PULSE# NODE_P NODE_N DC     VALUE  PULSE( VINIT  VPULSE TDELAY TRISE  TFALL  PWIDTH PERIOD )
VIN       VIN    0      DC     0      PULSE( 0      5      1n     1n     1n     50u     100u   )
XRE       VIN    VRE    RISE_E
XFE       VIN    VFE    FALL_E

.control 
*TRAN     TSTEP  TSTOP  TSTART TMAX   
tran      1u     3m     0       1u
set       pensize = 2
plot      vin vre vfe
plot       vfe
.endc

*=========Rising_Edge=========================
.SUBCKT   RISE_E VIN    OUT  
 R1       VIN    VTD    1meg
 C1       VTD    0      1p 
 BRE      OUT    0      V = 5*u(V(VIN)-V(VTD)-.1) 
.ENDS     RISE_E
*  ___                         ___
* |VIN|_/\  /\  /\_ VTD    ___|OUT|
* |___|   \/  \/  _|_    _|_  |___|
*              C1 ___   /BRE\
*       R1         |    \___/
*                 _|_    _|_
*                 ///    ///
*==============================================

*=========Falling_Edge=========================
.SUBCKT   FALL_E VIN    OUT  
 R1       VIN    VTD    1meg
 C1       VTD    0      1p 
 BRE      OUT    0      V = 5*u(V(VTD) -V(VIN)-.1) 
.ENDS     FALL_E
*  ___                         ___
* |VIN|_/\  /\  /\_ VTD    ___|OUT|
* |___|   \/  \/  _|_    _|_  |___|
*              C1 ___   /BRE\
*       R1         |    \___/
*                 _|_    _|_
*                 ///    ///
*==============================================

.end

